

Array-ifying a Sketch

Be careful not to sweep up too much into one array loop –

- add a separate array loop around each distinct step that applies to all of the things
- do *not* include setting or updating non-array animation variables in an array loop

Array-ifying a Sketch

each ellipse may have a different, unrelated y coordinate so this animation variable becomes an array variable

```

float y;
void setup () {
  size(400,400);
  y = random(0,height);
}

void draw () {
  background(255);
  fill(255,0,0);
  ellipse(width/2,y,20,20);
  y = y+1;
}
    
```

add array loops around each distinct step – initialize, draw, update

```

float[] y;
void setup () {
  size(400,400);
  y = new float[10];
  for ( int i = 0; i < y.length; i = i+1 ) {
    y[i] = random(0,height);
  }
}

void draw () {
  background(255);
  for ( int i = 0; i < y.length; i = i+1 ) {
    fill(255,0,0);
    ellipse(width/2,y[i],20,20);
  }
  for ( int i = 0; i < y.length; i = i+1 ) {
    y[i] = y[i]+1;
  }
}
    
```

CPSC 120: PIR 15

Array-ifying a Sketch – Example 2

```

float x, xspeed; // position and speed of ellipse

void setup () {
  size(400, 400);

  x = 10;
  xspeed = random(0, 5);
}

void draw () {
  background(255);
  ellipseMode(CENTER);

  // draw ellipse
  fill(255, 0, 0);
  ellipse(x, 10, 20, 20);

  // move ellipse
  x = x+xspeed;
}
    
```

```

// draw all the ellipses
for ( int i = 0; i < x.length; i = i+1 ) {
  fill(255, 0, 0);
  ellipse(x[i], 10, 20, 20);
}

// move all the ellipses
for ( int i = 0; i < x.length; i = i+1 ) {
  x[i] = x[i]+xspeed[i];
}
    
```

- two distinct steps applying to all the things – draw and move → separate array loop around each
 - original sketch had two steps – draw ellipse, move ellipse
 - array-ified sketch has two steps – draw all ellipses, move all ellipses

Array-ifying a Sketch – Example 2

```

float x, xspeed; // position and speed of ellipse

void setup () {
  size(400, 400);

  x = 10;
  xspeed = random(0, 5);
}

void draw () {
  background(255);
  ellipseMode(CENTER);

  // draw ellipse
  fill(255, 0, 0);
  ellipse(x, 10, 20, 20);

  // move ellipse
  x = x+xspeed;
}
    
```

```

// every ellipse starts with the same x coordinate
x = new float[20];
for ( int i = 0; i < x.length; i = i+1 ) {
  x[i] = 10;
}

// every ellipse has a random speed
xspeed = new float[20];
for ( int i = 0; i < xspeed.length; i = i+1 ) {
  xspeed[i] = random(0, 5);
}

// every ellipse starts with the same x coordinate
// and a random speed
x = new float[20];
xspeed = new float[20];
for ( int i = 0; i < x.length; i = i+1 ) {
  x[i] = 10;
  xspeed[i] = random(0, 5);
}
    
```

- initialization can be viewed as several steps – initialize x, initialize xspeed – or as one step – initialize ellipse
 - can have a separate array loop to initialize each array or a single array loop to initialize all of the arrays

Array-ifying a Sketch – Example 3

```

float x, xspeed; // position and speed of ellipse
float red; // color of ellipse

void setup () {
  size(400, 400);

  x = 10;
  xspeed = random(0, 5);
  red = 255;
}

void draw () {
  background(255);
  ellipseMode(CENTER);

  // draw ellipse
  fill(red, 0, 0);
  ellipse(x, 10, 20, 20);

  // move ellipse and update color
  if ( red > 0 ) {
    x = x+xspeed;
    red = red-1;
  }
}
    
```

```

float[] x, xspeed; // position and speed of ellipses
float red; // color of ellipses

void setup () {
  size(400, 400);

  // every ellipse starts with the same x coordinate
  // and a random speed
  x = new float[20];
  xspeed = new float[20];
  for ( int i = 0; i < x.length; i = i+1 ) {
    x[i] = 10;
    xspeed[i] = random(0, 5);
  }

  // all ellipses start out red
  red = 255;
}
    
```

- same color for all ellipses → red does not become an array variable and does not go into the initialization array loop

Array-ifying a Sketch – Example 3

```
float x, xspeed; // position and speed of ellipse
float red; // color of ellipse

void setup () {
  size(400, 400);
  x = 10;
  xspeed = random(0, 5);
  red = 255;
}

void draw () {
  background(255);
  ellipseMode(CENTER);
  // draw ellipse
  fill(red, 0, 0);
  ellipse(x, 10, 20, 20);
  // move ellipse and update color if not black
  if ( red > 0 ) {
    x = x+xspeed;
    red = red-1;
  }
}

// move all the ellipses and update color if not black
// *** INCORRECT!
for ( int i = 0; i < x.length; i = i+1 ) {
  if ( red > 0 ) {
    x[i] = x[i]+xspeed[i];
    red = red-1;
  }
}

// move all the ellipses and update color if not black
if ( red > 0 ) {
  for ( int i = 0; i < x.length; i = i+1 ) {
    x[i] = x[i]+xspeed[i];
  }
  red = red-1;
}

if ( red > 0 ) {
  // move ellipse
  // update color
}

if ( red > 0 ) {
  // move all ellipses
  // update color
}
```

- same color for all ellipses → update of red cannot go into an array loop